BasicIntroduction

October 23, 2016

0.1 A Basic Introduction to Julia

This quick introduction assumes that you have basic knowledge of some scripting language and provides an example of the Julia syntax. So before we explain anything, let's just treat it like a scripting language, take a head-first dive into Julia, and see what happens.

You'll notice that, given the right syntax, almost everything will "just work". There will be some peculiarities, and these we will be the facts which we will study in much more depth. Usually, these oddies/differences from other scripting languages are "the source of Julia's power".

0.1.1 Problems

Time to start using your noggin. Scattered in this document are problems for you to solve using Julia. Many of the details for solving these problems have been covered, some have not. You may need to use some external resources:

http://docs.julialang.org/en/release-0.5/manual/

https://gitter.im/JuliaLang/julia

Solve as many or as few problems as you can during these times. Please work at your own pace, or with others if that's how you're comfortable!

0.2 Documentation and "Hunting"

The main source of information is the Julia Documentation. Julia also provides lots of built-in documentation and ways to find out what's going on. The number of tools for "hunting down what's going on / available" is too numerous to explain in full detail here, so instead this will just touch on what's important. For example, the ? gets you to the documentation for a type, function, etc.

```
In [23]: ?copy
```

```
search: copy copy! copysign deepcopy unsafe_copy! cospi complex Complex
```

```
Out[23]:
```

```
copy(x)
```

Create a shallow copy of `x`: the outer structure is copied, but not all internal values. For example, copying an array produces a new array with identically-same elements as the original.

To find out what methods are available, we can use the `methods` function. For example, let's see how + is defined:

```
In [25]: methods(+)

Out[25]: # 163 methods for generic function "+":
         +(x::Bool, z::Complex{Bool}) at complex.jl:136
         +(x::Bool, y::Bool) at bool.jl:48
         +(x::Bool) at bool.jl:45
         +{T<:AbstractFloat}(x::Bool, y::T) at bool.jl:55
         +(x::Bool, z::Complex) at complex.jl:143
         +(x::Bool, A::AbstractArray{Bool,N<:Any}) at arraymath.jl:91
         +(x::Float32, y::Float32) at float.jl:239
         +(x::Float64, y::Float64) at float.jl:240
         +(z::Complex{Bool}, x::Bool) at complex.jl:137
         +(z::Complex{Bool}, x::Real) at complex.jl:151
         +(a::Float16, b::Float16) at float16.jl:136
         +(x::Char, y::Integer) at char.jl:40
         +(c::BigInt, x::BigFloat) at mpfr.jl:240
         +(a::BigInt, b::BigInt, c::BigInt, d::BigInt, e::BigInt) at gmp.jl:298
         +(a::BigInt, b::BigInt, c::BigInt, d::BigInt) at gmp.jl:291
         +(a::BigInt, b::BigInt, c::BigInt) at gmp.jl:285
         +(x::BigInt, y::BigInt) at gmp.jl:255
         +(x::BigInt, c::Union{UInt16,UInt32,UInt64,UInt8}) at gmp.jl:310
         +(x::BigInt, c::Union{Int16,Int32,Int64,Int8}) at gmp.jl:326
         +(a::BigFloat, b::BigFloat, c::BigFloat, d::BigFloat, e::BigFloat) at mpfr
         +(a::BigFloat, b::BigFloat, c::BigFloat, d::BigFloat) at mpfr.jl:381
         +(a::BigFloat, b::BigFloat, c::BigFloat) at mpfr.jl:375
         +(x::BigFloat, c::BigInt) at mpfr.jl:236
         +(x::BigFloat, y::BigFloat) at mpfr.jl:205
         +(x::BigFloat, c::Union{UInt16,UInt32,UInt64,UInt8}) at mpfr.jl:212
         +(x::BigFloat, c::Union{Int16,Int32,Int64,Int8}) at mpfr.jl:220
         +(x::BigFloat, c::Union{Float16,Float32,Float64}) at mpfr.jl:228
         +{T}(B::BitArray{2}, J::UniformScaling{T}) at linalg/uniformscaling.jl:38
         +(a::Base.Pkg.Resolve.VersionWeights.VWPreBuildItem, b::Base.Pkg.Resolve.V
         +(a::Base.Pkg.Resolve.VersionWeights.VWPreBuild, b::Base.Pkg.Resolve.Versi
         +(a::Base.Pkg.Resolve.VersionWeights.VersionWeight, b::Base.Pkg.Resolve.Ve
         +(a::Base.Pkg.Resolve.MaxSum.FieldValues.FieldValue, b::Base.Pkg.Resolve.M
         +(x::Base.Dates.CompoundPeriod, y::Base.Dates.CompoundPeriod) at dates/per
         +(x::Base.Dates.CompoundPeriod, y::Base.Dates.Period) at dates/periods.jl:
         +(x::Base.Dates.CompoundPeriod, y::Base.Dates.TimeType) at dates/periods.j
         +(dt::DateTime, z::Base.Dates.Month) at dates/arithmetic.jl:37
         +(dt::DateTime, y::Base.Dates.Year) at dates/arithmetic.jl:13
         +(x::DateTime, y::Base.Dates.Period) at dates/arithmetic.jl:64
         +(x::Date, y::Base.Dates.Day) at dates/arithmetic.jl:62
         +(x::Date, y::Base.Dates.Week) at dates/arithmetic.jl:60
         +(dt::Date, z::Base.Dates.Month) at dates/arithmetic.jl:43
         +(dt::Date, y::Base.Dates.Year) at dates/arithmetic.jl:17
         +(y::AbstractFloat, x::Bool) at bool.jl:57
         +{T<:Union{Int128,Int16,Int32,Int64,Int8,UInt128,UInt16,UInt32,UInt64,UInt
         +(x::Integer, y::Ptr) at pointer.jl:108
```

```
+(z::Complex, w::Complex) at complex.jl:125
+(z::Complex, x::Bool) at complex.jl:144
+(x::Real, z::Complex{Bool}) at complex.jl:150
+(x::Real, z::Complex) at complex.jl:162
+(z::Complex, x::Real) at complex.jl:163
+(x::Rational, y::Rational) at rational.jl:199
+(x::Integer, y::Char) at char.jl:41
+{N}(i::Integer, index::CartesianIndex{N}) at multidimensional.jl:58
+(c::Union{UInt16,UInt32,UInt64,UInt8}, x::BigInt) at gmp.jl:314
+(c::Union{Int16,Int32,Int64,Int8}, x::BigInt) at gmp.jl:327
+(c::Union{UInt16,UInt32,UInt64,UInt8}, x::BigFloat) at mpfr.jl:216
+(c::Union{Int16,Int32,Int64,Int8}, x::BigFloat) at mpfr.jl:224
+(c::Union{Float16,Float32,Float64}, x::BigFloat) at mpfr.jl:232
+(x::Irrational, y::Irrational) at irrationals.jl:88
+(x::Number) at operators.jl:115
+{T<:Number}(x::T, y::T) at promotion.jl:255
+(x::Number, y::Number) at promotion.jl:190
+(r1::OrdinalRange, r2::OrdinalRange) at operators.jl:505
+{T<:AbstractFloat}(r1::FloatRange{T}, r2::FloatRange{T}) at operators.jl:
+{T<:AbstractFloat}(r1::LinSpace{T}, r2::LinSpace{T}) at operators.jl:531
+(r1::Union{FloatRange,LinSpace,OrdinalRange}, r2::Union{FloatRange,LinSpa
+(x::Ptr, y::Integer) at pointer.jl:106
+(A::BitArray, B::BitArray) at bitarray.jl:1042
+(A::Array{T<:Any,2}, B::Diagonal) at linalg/special.jl:121
+(A::Array{T<:Any,2}, B::Bidiagonal) at linalg/special.jl:121
+(A::Array{T<:Any,2}, B::Tridiagonal) at linalg/special.jl:121
+(A::Array{T<:Any,2}, B::SymTridiagonal) at linalg/special.jl:130
+(A::Array{T<:Any,2}, B::Base.LinAlg.AbstractTriangular) at linalg/special
+(A::Array, B::SparseMatrixCSC) at sparse/sparsematrix.jl:1711
+{P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Period}}(x::Union{Base.Res
+(A::AbstractArray{Bool,N<:Any}, x::Bool) at arraymath.jl:90
+(A::SymTridiagonal, B::SymTridiagonal) at linalg/tridiag.jl:96
+(A::Tridiagonal, B::Tridiagonal) at linalg/tridiag.jl:494
+(A::UpperTriangular, B::UpperTriangular) at linalg/triangular.jl:357
+(A::LowerTriangular, B::LowerTriangular) at linalg/triangular.jl:358
+(A::UpperTriangular, B::Base.LinAlg.UnitUpperTriangular) at linalg/triang
+(A::LowerTriangular, B::Base.LinAlg.UnitLowerTriangular) at linalg/triang
+(A::Base.LinAlg.UnitUpperTriangular, B::UpperTriangular) at linalg/triang
+(A::Base.LinAlg.UnitLowerTriangular, B::LowerTriangular) at linalg/triang
+(A::Base.LinAlg.UnitUpperTriangular, B::Base.LinAlg.UnitUpperTriangular)
+(A::Base.LinAlg.UnitLowerTriangular, B::Base.LinAlg.UnitLowerTriangular)
+(A::Base.LinAlg.AbstractTriangular, B::Base.LinAlg.AbstractTriangular) at
+(Da::Diagonal, Db::Diagonal) at linalg/diagonal.jl:110
+(A::Bidiagonal, B::Bidiagonal) at linalg/bidiag.jl:256
+(UL::UpperTriangular, J::UniformScaling) at linalg/uniformscaling.jl:55
+(UL::Base.LinAlg.UnitUpperTriangular, J::UniformScaling) at linalg/unifor
+(UL::LowerTriangular, J::UniformScaling) at linalg/uniformscaling.jl:55
+(UL::Base.LinAlg.UnitLowerTriangular, J::UniformScaling) at linalg/unifor
```

```
+(A::Diagonal, B::Bidiagonal) at linalg/special.jl:120
+(A::Bidiagonal, B::Diagonal) at linalg/special.jl:121
+(A::Diagonal, B::Tridiagonal) at linalg/special.jl:120
+(A::Tridiagonal, B::Diagonal) at linalg/special.jl:121
+(A::Diagonal, B::Array{T<:Any,2}) at linalg/special.jl:120
+(A::Bidiagonal, B::Tridiagonal) at linalg/special.jl:120
+(A::Tridiagonal, B::Bidiagonal) at linalg/special.jl:121
+(A::Bidiagonal, B::Array{T<:Any,2}) at linalg/special.jl:120
+(A::Tridiagonal, B::Array{T<:Any,2}) at linalg/special.jl:120
+(A::SymTridiagonal, B::Tridiagonal) at linalg/special.jl:129
+(A::Tridiagonal, B::SymTridiagonal) at linalg/special.jl:130
+(A::SymTridiagonal, B::Array{T<:Any,2}) at linalg/special.jl:129
+(A::Diagonal, B::SymTridiagonal) at linalg/special.jl:138
+(A::SymTridiagonal, B::Diagonal) at linalg/special.jl:139
+(A::Bidiagonal, B::SymTridiagonal) at linalg/special.jl:138
+(A::SymTridiagonal, B::Bidiagonal) at linalg/special.jl:139
+(A::Diagonal, B::UpperTriangular) at linalg/special.jl:150
+(A::UpperTriangular, B::Diagonal) at linalg/special.jl:151
+(A::Diagonal, B::Base.LinAlg.UnitUpperTriangular) at linalg/special.jl:15
+(A::Base.LinAlg.UnitUpperTriangular, B::Diagonal) at linalg/special.jl:15
+(A::Diagonal, B::LowerTriangular) at linalg/special.jl:150
+(A::LowerTriangular, B::Diagonal) at linalg/special.jl:151
+(A::Diagonal, B::Base.LinAlg.UnitLowerTriangular) at linalg/special.jl:15
+(A::Base.LinAlg.UnitLowerTriangular, B::Diagonal) at linalg/special.jl:15
+(A::Base.LinAlg.AbstractTriangular, B::SymTridiagonal) at linalg/special.
+(A::SymTridiagonal, B::Base.LinAlg.AbstractTriangular) at linalg/special.
+(A::Base.LinAlg.AbstractTriangular, B::Tridiagonal) at linalg/special.jl:
+(A::Tridiagonal, B::Base.LinAlg.AbstractTriangular) at linalg/special.jl:
+(A::Base.LinAlg.AbstractTriangular, B::Bidiagonal) at linalg/special.jl:1
+(A::Bidiagonal, B::Base.LinAlg.AbstractTriangular) at linalg/special.jl:1
+(A::Base.LinAlg.AbstractTriangular, B::Array{T<:Any,2}) at linalg/special
+{P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Period}}(Y::Union{Base.Res
+{P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Period},Q<:Union{Base.Date
+{T<:Base.Dates.TimeType,P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Per
+{T<:Base.Dates.TimeType}(r::Range{T}, x::Base.Dates.Period) at dates/rang
+{Tv1,Ti1,Tv2,Ti2}(A_1::SparseMatrixCSC{Tv1,Ti1}, A_2::SparseMatrixCSC{Tv2
+(A::SparseMatrixCSC, B::Array) at sparse/sparsematrix.jl:1709
+(A::SparseMatrixCSC, J::UniformScaling) at sparse/sparsematrix.jl:3811
+(x::AbstractSparseArray{Tv<:Any,Ti<:Any,1}, y::AbstractSparseArray{Tv<:An
+(x::Union{Base.ReshapedArray{T<:Any,1,A<:DenseArray,MI<:Tuple{Vararg{Base
+(x::AbstractSparseArray{Tv<:Any,Ti<:Any,1}, y::Union{Base.ReshapedArray{T
+{T<:Number}(x::AbstractArray{T,N<:Any}) at abstractarraymath.jl:91
+{R,S}(A::AbstractArray{R,N<:Any}, B::AbstractArray{S,N<:Any}) at arraymat
+(A::AbstractArray, x::Number) at arraymath.jl:94
+(x::Number, A::AbstractArray) at arraymath.jl:95
+{N}(index1::CartesianIndex{N}, index2::CartesianIndex{N}) at multidimensi
+{N}(index::CartesianIndex{N}, i::Integer) at multidimensional.jl:57
+(J1::UniformScaling, J2::UniformScaling) at linalg/uniformscaling.jl:37
```

```
+(J::UniformScaling, B::BitArray{2}) at linalg/uniformscaling.jl:39
+(J::UniformScaling, A::AbstractArray{T<:Any,2}) at linalg/uniformscaling.
+(J::UniformScaling, x::Number) at linalg/uniformscaling.jl:41
+(x::Number, J::UniformScaling) at linalg/uniformscaling.jl:42
+{TA,TJ}(A::AbstractArray{TA,2}, J::UniformScaling{TJ}) at linalg/uniforms
+{T}(a::Base.Pkg.Resolve.VersionWeights.HierarchicalValue{T}, b::Base.Pkg.
+{P<:Base.Dates.Period}(x::P, y::P) at dates/periods.jl:70
+(x::Base.Dates.Period, y::Base.Dates.Period) at dates/periods.jl:311
+(y::Base.Dates.Period, x::Base.Dates.CompoundPeriod) at dates/periods.jl:
+(y::Base.Dates.Period, x::Base.Dates.TimeType) at dates/arithmetic.jl:66
+{T<:Base.Dates.TimeType}(x::Base.Dates.Period, r::Range{T}) at dates/rang
+(x::Union{Base.Dates.CompoundPeriod,Base.Dates.Period}) at dates/periods.
+{P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Period}}(x::Union{Base.Dat
+(x::Base.Dates.TimeType) at dates/arithmetic.jl:8
+(a::Base.Dates.TimeType, b::Base.Dates.Period, c::Base.Dates.Period) at d
+(a::Base.Dates.TimeType, b::Base.Dates.Period, c::Base.Dates.Period, d::B
+(x::Base.Dates.TimeType, y::Base.Dates.CompoundPeriod) at dates/periods.j
+(x::Base.Dates.Instant) at dates/arithmetic.jl:4
+{T<:Base.Dates.TimeType}(x::AbstractArray{T,N<:Any}, y::Union{Base.Dates.
+{T<:Base.Dates.TimeType}(y::Union{Base.Dates.CompoundPeriod,Base.Dates.Pe
+{P<:Union{Base.Dates.CompoundPeriod,Base.Dates.Period}}(y::Base.Dates.Tim
+(a, b, c, xs...) at operators.jl:138
```

We can inspect a type by finding its fields with `fieldnames`

```
In [40]: fieldnames(LinSpace)

Out[40]: 4-element Array{Symbol,1}:
          :start
          :stop
          :len
          :divisor
```

and find out which method was used with the `@which` macro:

```
In [43]: @which copy([1,2,3])

Out[43]: copy{T<:Array{T,N}}(a::T) at array.jl:70
```

Notice that this gives you a link to the source code where the function is defined.
Lastly, we can find out what type a variable is with the `typeof` function:

```
In [44]: a = [1;2;3]
         typeof(a)

Out[44]: Array{Int64,1}
```

0.2.1 Array Syntax

The array syntax is similar to MATLAB's conventions.

```
In [11]: a = Vector{Float64}(5) # Create a length 5 Vector (dimension 1 array) of F

         a = [1;2;3;4;5] # Create the column vector [1 2 3 4 5]

         a = [1 2 3 4] # Create the row vector [1 2 3 4]

         a[3] = 2 # Change the third element of a (using linear indexing) to 2

         b = Matrix{Float64}(4,2) # Define a Matrix of Float64's of size (4,2)

         c = Array{Float64,4}(4,5,6,7) # Define a (4,5,6,7) array of Float64's

         mat    = [1 2 3 4
                   3 4 5 6
                   4 4 4 6
                   3 3 3 3] #Define the matrix inline

         mat[1,2] = 4 # Set element (1,2) (row 1, column 2) to 4

         mat

Out[11]: 4×4 Array{Int64,2}:
          1  4  3  4
          3  4  5  6
          4  4  4  6
          3  3  3  3
```

Note that, in the console (called the REPL), you can use `;` to surpress the output. In a script this is done automatically. Note that the "value" of an array is its pointer to the memory location. This means that arrays which are set equal affect the same values:

```
In [12]: a = [1;3;4]
         b = a
         b[1] = 10
         a

Out[12]: 3-element Array{Int64,1}:
          10
           3
           4
```

To set an array equal to the values to another array, use copy

```
In [13]: a = [1;4;5]
         b = copy(a)
         b[1] = 10
         a
```

```
Out[13]: 3-element Array{Int64,1}:
          1
          4
          5
```

We can also make an array of a similar size and shape via the function `similar`, or make an array of zeros/ones with `zeros` or `ones` respectively:

```
In [14]: c = similar(a)
         d = zeros(a)
         e = ones(a)
         println(c); println(d); println(e)

[786432,0,0]
[0,0,0]
[1,1,1]
```

Note that arrays can be index'd by arrays:

```
In [15]: a[1:2]

Out[15]: 2-element Array{Int64,1}:
          1
          4
```

Arrays can be of any type, specified by the type parameter. One interesting thing is that this means that arrays can be of arrays:

```
In [12]: a = Vector{Vector{Float64}}(3)
         a[1] = [1;2;3]
         a[2] = [1;2]
         a[3] = [3;4;5]
         a

Out[12]: 3-element Array{Array{Float64,1},1}:
          [1.0,2.0,3.0]
          [1.0,2.0]
          [3.0,4.0,5.0]
```

Question 1 Can you explain the following behavior? Julia's community values consistancy of the rules, so all of the behavior is deducible from simple rules. (Hint: I have noted all of the rules involved here).

```
In [14]: b = a
         b[1] = [1;4;5]
         a
```

```
Out[14]: 3-element Array{Array{Float64,1},1}:
          [1.0,4.0,5.0]
          [1.0,2.0]
          [3.0,4.0,5.0]
```

To fix this, there is a recursive copy function: `deepcopy`

```
In [16]: b = deepcopy(a)
         b[1] = [1;2;3]
         a

Out[16]: 3-element Array{Array{Float64,1},1}:
          [1.0,4.0,5.0]
          [1.0,2.0]
          [3.0,4.0,5.0]
```

For high performance, Julia provides mutating functions. These functions change the input values that are passed in, instead of returning a new value. By convention, mutating functions tend to be defined with a `!` at the end and tend to mutate their first argument. An example of a mutating function in `scale!` which scales an array by a scalar (or array)

```
In [19]: a = [1;6;8]
         scale!(a,2) # a changes

Out[19]: 3-element Array{Int64,1}:
           2
          12
          16
```

The purpose of mutating functions is that they allow one to reduce the number of memory allocations which is crucial for achiving high performance.

0.3 Control Flow

Control flow in Julia is pretty standard. You have your basic for and while loops, and your if statements. There's more in the documentation.

```
In [16]: for i=1:5 #for i goes from 1 to 5
             println(i)
         end

         t = 0
         while t<5
             println(t)
             t+=1 # t = t + 1
         end
```

```
school = :UCI

if school==:UCI
    println("ZotZotZot")
else
    println("Not even worth discussing.")
end
```

```
1
2
3
4
5
0
1
2
3
4
ZotZotZot
```

One interesting feature about Julia control flow is that we can write multiple loops in one line:

In [17]:
```
for i=1:2,j=2:4
    println(i*j)
end
```

```
2
3
4
4
6
8
```

Problem 1 Use Julia's array and control flow syntax in order to define the NxN Strang matrix:

$$\begin{bmatrix} -2 & 1 & & & \\ 1 & -2 & 1 & & \\ & \ddots & \ddots & \ddots & \\ & & \ddots & \ddots & 1 \\ & & & 1 & -2 \end{bmatrix}$$

In [ ]:
```
#### Prepare Data

X = rand(1000, 3)                # feature matrix
a0 = rand(3)                     # ground truths
y = X * a0 + 0.1 * randn(1000);  # generate response
```

Problem 2 Given an Nx3 array of data (`randn(N,3)`) and a Nx1 array of outcomes, produce the data matrix `X` which appends a column of 1's to the data matrix, and solve for the 4x1 array β via β`X = b` using `qrfact` or `\`. (Note: This is linear regression)

Problem 3 Compare your results to that of using `llsq` from `MultivariateStats.jl` (note: you need to go find the documentation to find out how to use this!)

Problem 4 Compare your results to that of using ordinary least squares regression from `GLM.jl`

Problem 5 The logistic difference equation is defined by the recursion

$$b_{n+1} = r * b_n(1 - b_n)$$

where b_n is the number of bunnies at time n. Starting with $b_0 = .25$, by around 400 iterations this will reach a steady state. This steady state (or steady periodic state) is dependent on r. Write a function which solves for the steady state(s) for each given r, and plot "every state" in the steady attractor for each r (x-axis is r, y=value seen in the attractor) using Plots.jl. Take $r \in (2.9, 4)$

0.4 Function Syntax

```
In [1]: f(x,y) = 2x+y # Create an inline function

Out[1]: f (generic function with 1 method)

In [22]: f(1,2) # Call the function

Out[22]: 4

In [2]: function f(x)
          x+2
        end # Long form definition

Out[2]: f (generic function with 2 methods)
```

By default, Julia functions return the last value computed within them.

```
In [26]: f(2)

Out[26]: 4
```

A key feature of Julia is multiple dispatch. Notice here that there is "one function", `f`, with two methods. Methods are the actionable parts of a function. Here, there is one method defined as `(::Any,::Any)` and `(::Any)`, meaning that if you give `f` two values then it will call the first method, and if you give it one value then it will call the second method.

Multiple dispatch works on types. To define a dispatch on a type, use the `::Type` signifier:

```
In [3]: f(x::Int,y::Int) = 3x+2y

Out[3]: f (generic function with 3 methods)
```

Julia will dispatch onto the strictest acceptible type signature.

```
In [30]: f(2,3) # 3x+2y

Out[30]: 12

In [32]: f(2.0,3) # 2x+y since 2.0 is not an Int

Out[32]: 7.0
```

Types in signatures can be parametric. For example, we can define a method for "two values are passed in, both Numbers and having the same type". Note that `<:` means "a subtype of".

```
In [4]: f{T<:Number}(x::T,y::T) = 4x+10y

Out[4]: f (generic function with 4 methods)

In [37]: f(2,3) # 3x+2y since (::Int,::Int) is stricter

Out[37]: 12

In [4]: f(2.0,3.0) # 4x+10y

Out[4]: 38.0
```

Note that type parameterizations can have as many types as possible, and do not need to declare a supertype. For example, we can say that there is an `x` which must be a Number, while `y` and `z` must match types:

```
In [5]: f{T<:Number,T2}(x::T,y::T2,z::T2) = 5x + 5y + 5z

Out[5]: f (generic function with 5 methods)
```

We will go into more depth on multiple dispatch later since this is the core design feature of Julia. The key feature is that Julia functions specialize on the types of their arguments. This means that `f` is a separately compiled function for each method (and for parametric types, each possible method). The first time it is called it will compile.

Question 2 Can you explain these timings?

```
In [6]: f(x,y,z,w) = x+y+z+w
        @time f(1,1,1,1)
        @time f(1,1,1,1)
        @time f(1,1,1,1)
        @time f(1,1,1,1.0)
        @time f(1,1,1,1.0)
```

```
  0.003880 seconds (387 allocations: 21.150 KB)
  0.000002 seconds (4 allocations: 160 bytes)
  0.000001 seconds (3 allocations: 144 bytes)
  0.002570 seconds (1.33 k allocations: 69.866 KB)
  0.000001 seconds (4 allocations: 160 bytes)


Out[6]: 4.0
```

Note that functions can also feature optional arguments:

```
In [42]: function test_function(x,y;z=0) #z is an optional argument
           if z==0
             return x+y,x*y #Return a tuple
           else
           return x*y*z,x+y+z #Return a different tuple
           #whitespace is optional
           end #End if statement
         end #End function definition

Out[42]: test_function (generic function with 1 method)
```

Here, if z is not specified, then it's 0.

```
In [45]: x,y = test_function(1,2)

Out[45]: (3,2)

In [46]: x,y = test_function(1,2;z=3)

Out[46]: (6,6)
```

Notice that we also featured multiple return values.

```
In [47]: println(x); println(y)

6
6
```

The return type for multiple return values is a Tuple. The syntax for a tuple is `(x,y,z,...)` or inside of functions you can use the shorthand `x,y,z,...` as shown.

Note that functions in Julia are "first-class". This means that functions are just a type themselves. Therefore functions can make functions, you can store functions as variables, pass them as variables, etc. For example:

```
In [6]: function function_playtime(x) #z is an optional argument
            y = 2+x
            function test()
                2y # y is defined in the previous scope, so it's available here
            end
            z = test() * test()
            return z,test
        end #End function definition
        z,test = function_playtime(2)

Out[6]: (64,test)

In [14]: test()

Out[14]: 8
```

Notice that `test()` does not get passed in `y` but knows what `y` is. This is due to the function scoping rules: an inner function can know the variables defined in the same scope as the function. This rule is recursive, leading us to the conclusion that the top level scope is global. Yes, that means

```
In [18]: a = 2

Out[18]: 2
```

defines a global variable. We will go into more detail on this.

Lastly we show the anonymous function syntax. This allows you to define a function inline.

```
In [20]: g = (x,y) -> 2x+y

Out[20]: (::#5) (generic function with 1 method)
```

Unlike named functions, `g` is simply a function in a variable and can be overwritten at any time:

```
In [21]: g = (x) -> 2x

Out[21]: (::#7) (generic function with 1 method)
```

An anonymous function cannot have more than 1 dispatch. However, as of v0.5, they are compiled and thus do not have any performance disadvantages from named functions.

0.5 Type Declaration Syntax

A type is what in many other languages is an "object". If that is a foreign concept, thing of a type as a thing which has named components. A type is the idea for what the thing is, while an instantiation of the type is a specific one. For example, you can think of a car as having an make and a model. So that means a Toyota RAV4 is an instantiation of the car type.

In Julia, we would define the car type as follows:

```
In [48]: type Car
             make
             model
         end
```

We could then make the instance of a car as follows:

```
In [49]: mycar = Car("Toyota","Rav4")

Out[49]: Car("Toyota","Rav4")
```

Here I introduced the string syntax for Julia which uses "..." (like most other languages, I'm glaring at you MATLAB). I can grab the "fields" of my type using the . syntax:

```
In [51]: mycar.make

Out[51]: "Toyota"
```

To "enhance Julia's performance", one usually likes to make the typing stricter. For example, we can define a WorkshopParticipant (notice the convention for types is capital letters, CamelCase) as having a name and a field. The name will be a string and the field will be a Symbol type, (defined by :Symbol, which we will go into plenty more detail later).

```
In [52]: type WorkshopParticipant
             name::String
             field::Symbol
         end
         tony = WorkshopParticipant("Tony",:physics)

Out[52]: WorkshopParticipant("Tony",:physics)
```

As with functions, types can be set "parametrically". For example, we can have an StaffMember have a name and a field, but also an age. We can allow this age to be any Number type as follows:

```
In [1]: type StaffMember{T<:Number}
            name::String
            field::Symbol
            age::T
        end
        ter = StaffMember("Terry",:football,17)

Out[1]: StaffMember{Int64}("Terry",:football,17)
```

The rules for parametric typing is the same as for functions. Note that most of Julia's types, like Float64 and Int, are natively defined in Julia in this manner. This means that there's no limit for user defined types, only your imagination. Indeed, many of Julia's features first start out as a prototyping package before it's ever moved into Base (the Julia library that ships as the Base module in every installation).

Lastly, there exist abstract types. These types cannot be instantiated but are used to build the type hierarchy. You've already seen one abstract type, Number. We can define one for Person using the Abstract keyword

```
In [15]: abstract Person
```

Then we can set types as a subtype of person

```
In [16]: type Student <: Person
             name
             grade
         end
```

You can define type heirarchies on abstract types. See the beautiful explanation at: http://docs.julialang.org/en/release-0.5/manual/types/#abstract-types

```
In [ ]: abstract AbstractStudent <: AbstractPerson
```

Another "version" of type is `immutable`. When one uses `immutable`, the fields of the type cannot be changed. However, Julia will automatically stack allocate immutable types, whereas standard types are heap allocated. If this is unfamiliar terminology, then think of this as meaning that immutable types are able to be stored closer to the CPU and have less cost for memory access (this is a detail not present in many scripting languages). Many things like Julia's built-in Number types are defined as `immutable` in order to give good performance.

```
In [7]: immutable Field
            name
            school
        end
        ds = Field(:DataScience,[:PhysicalScience;:ComputerScience])
```

```
Out[7]: Field(:DataScience,Symbol[:PhysicalScience,:ComputerScience])
```

Question 3 Can you explain this interesting quirk? Thus Field is immutable, meaning that `ds.name` and `ds.school` cannot be changed:

```
In [64]: ds.name = :ComputationalStatistics
```

```
LoadError: type Field is immutable
while loading In[64], in expression starting on line 1
```

However, the following is allowed:

```
In [65]: push!(ds.school,:BiologicalScience)
         ds.school
```

```
Out[65]: 3-element Array{Symbol,1}:
          :PhysicalScience
          :ComputerScience
          :BiologicalScience
```

(Hint: recall that an array is not the values itself, but a pointer to the memory of the values)

One important detail in Julia is that everything is a type (and every piece of code is an Expression type, more on this later). Thus functions are also types, which we can access the fields of. Not only is everything compiled down to C, but all of the "C parts" are always accessible. For example, we can, if we so choose, get a function pointer:

```
In [46]: foo(x) = 2x
         first(methods(foo)).lambda_template.fptr

Out[46]: Ptr{Void} @0x0000000000000000
```

Problem 6 Make a function `person_info(x)` where, if `x` is a any type of person, print their name. However, if `x` is a student, print their name and their grade. Make a new type which is a graduate student, and have it print their name and grade as well. If `x` is anything else, throw an error. Do not using branching (`if`), use multiple dispatch to solve the problem!

Note that in order to do this you will need to re-structure the type hierarchy. Make an AbstractPerson and AbstractStudent type, define the subclassing structure, and write dispatches on these abstract types. Note that you cannot define subclasses of concrete types!

(Not only is the multiple-dispatch way more Julian, we will see later that it also has a lot of performance enhancements due to how it interacts with the compiler).

Problem 7 In mathematics, a matrix is known to be a linear operator. In many cases, this can have huge performance advantages because, if you know a function which "acts like a matrix" but does not form the matrix itself, you can save the time that it takes to allocate the matrix (sometimes the matrix may not fit in memory!)

Therefore, instead of solving regressions on matrices, let's be brave and generalize our regression algorithm to work on any operator, and make it solve the problem for the Laplacian operator. Here are the steps that are required to do this:

- Make an abstract type `AbstractOperator`
- Re-define AbstractMatrix as a subtype
- Define a concrete type `LaplacianOperator` which holds a function. This function `f(x)` should calculate the discrete Laplacian of `x` (i.e. multiply it on the left by the Strang matrix, but without forming the matrix!)
- Write dispatches for `*(::LaplacianOperator,::Vector)`, `eltype(::LaplacianOperator)`, `size(::LaplacianOperator,d::Integer)`.
- Write `least_square(::Operator,::Vector)` to solve the least-square approximation problem `Ax=b` for any operator. Note that since you do not have a matrix, you cannot use `\` or factorizations like `qrfact`. Instead, take a look at `cg` in IterativeSolvers.jl
- Test your least_square function vs `llsq` and `lm` (where you use the data matrix as a Strang matrix). Do you get the same result?

0.6 Some Basic Types

Julia provides many basic types. Indeed, you will come to know Julia as a system of multiple dispatch on types, meaning that the interaction of types with functions is core to the design.

0.6.1 Lazy Iterator Types

While MATLAB or Python has easy functions for building arrays, Julia tends to side-step the actual "array" part with specially made types. One such example are ranges. To define a range, use the `start:stepsize:end` syntax. For example:

```
In [45]: a = 1:5
         println(a)
         b = 1:2:10
         println(b)

1:5
1:2:9
```

We can use them like any array. For example:

```
In [47]: println(a[2]); println(b[3])

2
5
```

But what is `b`?

```
In [50]: println(typeof(b))

StepRange{Int64,Int64}
```

`b` isn't an array, it's a StepRange. A StepRange has the ability to act like an array using its fields:

```
In [52]: fieldnames(StepRange)

Out[52]: 3-element Array{Symbol,1}:
          :start
          :step
          :stop
```

Note that at any time we can get the array from these kinds of type via the `collect` function:

```
In [55]: c = collect(a)
```

```
Out[55]: 5-element Array{Int64,1}:
          1
          2
          3
          4
          5
```

The reason why lazy iterator types are preferred is that they do not do the computations until it's absolutely necessary, and they take up much less space. We can check this with `@time`:

```
In [7]: @time a = 1:100000
        @time a = 1:100
        @time b = collect(1:100000);

  0.000005 seconds (6 allocations: 240 bytes)
  0.000003 seconds (5 allocations: 192 bytes)
  0.007849 seconds (189 allocations: 792.844 KB)
```

Notice that the amount of time the range takes is much shorter. This is mostly because there is a lot less memory allocation needed: only a `StepRange` is built, and all that holds is the three numbers. However, `b` has to hold `100000` numbers, leading to the huge difference.

Problem 8 If you know `start`, `step`, and `stop`, how do you calculate the `i`th value? Can you create a function MyRange which where for `a` being a `MyRange`, and `a[i]` is the correct value? Use the Julia array interface in order to define the function for the `a[i]` syntax on your type.

Problem 9 Do ?linspace. Make your own LinSpace object using the array interface.
http://ucidatascienceinitiative.github.io/IntroToJulia/Html/ArrayIteratorInterfaces
Do your implementations obay dimensional analysis? Try using the package `Unitful` to build arrays of numbers with units (i.e. an array of numbers who have values of Newtons), and see if you can make your LinSpace not give errors.

Problem 10 Check your implementation vs the source code of Ranges.jl. Tim Holy is the master of Julia arrays, learn from him!

Problem 11 Check out the call overloading notebook:
http://ucidatascienceinitiative.github.io/IntroToJulia/Html/CallOverloading
Overload the call on the UnitStepRange to give an interpolated value at intermediate points, i.e. if `a=1:2:10`, then `a(1.5)=2`.

0.6.2 Dictionaries

Another common type is the Dictionary. It allows you to access (key,value) pairs in a named manner. For example:

```
In [1]: d = Dict(:test=>2,"silly"=>:suit)
        println(d[:test])
        println(d["silly"])
```

```
2
suit
```

0.6.3 Tuples

Tuples are immutable arrays. That means they can't be changed. However, they are super fast. They are made with the `(x,y,z,...)` syntax and are the standard return type of functions which return more than one object.

```
In [4]: tup = (2.,3) # Don't have to match types
        x,y = (3.0,"hi") # Can separate a tuple to multiple variables
```

```
Out[4]: (3.0,"hi")
```

0.7 Metaprogramming

Metaprogramming is a huge feature of Julia. The key idea is that every statement in Julia is of the type `Expression`. Julia operators by building an Abstract Syntax Tree (AST) from the Expressions. You've already been exposed to this a little bit: a `Symbol` (like `:PhysicalSciences` is not a string because it is part of the AST, and thus is part of the parsing/expression structure. One interesting thing is that symbol comparisons are O(1) while string comparisons, like always, are O(n)) is part of this, and macros (the weird functions with an `@`) are functions on expressions.

Thus you can think of metaprogramming as "code which takes in code and outputs code". One basic example is the `@time` macro:

```
In [85]: macro my_time(ex)
             return quote
               local t0 = time()
               local val = $ex
               local t1 = time()
               println("elapsed time: ", t1-t0, " seconds")
               val
             end
         end
```

```
        LoadError: error in method definition: function Base.@time must be explicit
    while loading In[85], in expression starting on line 1
```

This takes in an expression `ex`, gets the time before and after evaluation, and prints the elapsed time between (the real time macro also calculates the allocations as seen earlier). Note that `$ex` "interpolates" the expression into the macro. Going into detail on metaprogramming is a large step from standard scripting and will be a later session.

Why macros? One reason is because it lets you define any syntax you want. Since it operates on the expressions themselves, as long as you know how to parse the expression into working

code, you can "choose any syntax" to be your syntax. A case study will be shown later. Another reason is because these are done at "parse time" and those are only called once (before the function compilation).